
Abstracción y tipos abstractos de datos
I La abstracción es un mecanismo para comprender problemas

que involucran una gran cantidad de detalles.
I Aspectos de la abstracción:

I Destacar los detalles relevantes.
I Ocultar los detalles irrelevantes.

I Un tipo abstracto de datos (TAD) es una colección de valores
y operaciones que se definen mediante una especificación que es
independiente de cualquier representación.

I Un TAD es una abstracción:
I Se destacan los detalles (normalmente pocos) de la especificación

(el qué).
I Se ocultan los detalles (normalmente numerosos) de la

implementación (el cómo).
I Analogía con las estructuras algebraicas.

Descripción informal de las pilas

I Una pila es una estructura de datos, caracterizada por ser una
secuencia de elementos en la que las operaciones de inserción y
extracción se realizan por el mismo extremo.

I La pilas también se llaman estructuras LIFO (del inglés Last In
First Out), debido a que el último elemento en entrar será el
primero en salir.

I Analogía con las pilas de platos.

Signatura del TAD de las pilas

I Signatura:
vacia :: Pila a
apila :: a -> Pila a -> Pila a
cima :: Pila a -> a
desapila :: Pila a -> Pila a
esVacia :: Pila a -> Bool

I Descripción:
I vacia es la pila vacía.
I (apila x p) es la pila obtenida añadiendo x al principio de p.
I (cima p) es la cima de la pila p.
I (desapila p) es la pila obtenida suprimiendo la cima de p.
I (esVacia p) se verifica si p es la pila vacía.

Propiedades de las pilas

1. cima (apila x p) == x

2. desapila (apila x p) == p

3. esVacia vacia

4. not (esVacia (apila x p))

Las pilas mediante tipos de datos algebraicos
I Cabecera del módulo:

module PilaConTipoDeDatoAlgebraico
(Pila,
vacia, -- Pila a
apila, -- a -> Pila a -> Pila a
cima, -- Pila a -> a
desapila, -- Pila a -> Pila a
esVacia -- Pila a -> Bool

) where

I Tipo de dato algebraico de las pilas.

data Pila a = Vacia | P a (Pila a)
deriving Eq

Las pilas mediante tipos de datos algebraicos
I Procedimiento de escritura de pilas.

instance (Show a) => Show (Pila a) where
showsPrec p Vacia cad = showChar ’-’ cad
showsPrec p (P x s) cad =

shows x (showChar ’|’ (shows s cad))

I Ejemplo de pila:
I Definición

p1 :: Pila Int
p1 = apila 1 (apila 2 (apila 3 vacia))

I Sesión
ghci> p1
1|2|3|-

Las pilas mediante tipos de datos algebraicos
I vacia es la pila vacía. Por ejemplo,

ghci> vacia
-

vacia :: Pila a
vacia = Vacia

I (apila x p) es la pila obtenida añadiedo x encima de la pila p. Por
ejemplo,
apila 4 p1 => 4|1|2|3|-

apila :: a -> Pila a -> Pila a
apila x p = P x p

Las pilas mediante tipos de datos algebraicos
I (cima p) es la cima de la pila p. Por ejemplo,

cima p1 == 1

cima :: Pila a -> a
cima Vacia = error "cima: pila vacia"
cima (P x _) = x

I (desapila p) es la pila obtenida suprimiendo la cima de la pila
p. Por ejemplo,
desapila p1 => 2|3|-

desapila :: Pila a -> Pila a
desapila Vacia = error "desapila: pila vacia"
desapila (P _ p) = p

Las pilas mediante tipos de datos algebraicos

I (esVacia p) se verifica si p es la pila vacía. Por ejemplo,
esVacia p1 == False
esVacia vacia == True

esVacia :: Pila a -> Bool
esVacia Vacia = True
esVacia _ = False

Implementación de las pilas mediante listas
I Cabecera del módulo

module PilaConListas
(Pila,
vacia, -- Pila a
apila, -- a -> Pila a -> Pila a
cima, -- Pila a -> a
desapila, -- Pila a -> Pila a
esVacia -- Pila a -> Bool

) where

I Tipo de datos de las pilas:

newtype Pila a = P [a]
deriving Eq

Implementación de las pilas mediante listas

I Procedimiento de escritura de pilas.

instance (Show a) => Show (Pila a) where
showsPrec p (P []) cad = showChar ’-’ cad
showsPrec p (P (x:xs)) cad

= shows x (showChar ’|’ (shows (P xs) cad))

I Ejemplo de pila: p1 es la pila obtenida anadiéndole los elementos
3, 2 y 1 a la pila vacía. Por ejemplo,
ghci> p1
1|2|3|-

p1 = apila 1 (apila 2 (apila 3 vacia))

Implementación de las pilas mediante listas

I vacia es la pila vacía. Por ejemplo,
ghci> vacia
-

vacia :: Pila a
vacia = P []

I (apila x p) es la pila obtenida añadiendo x encima de la pila
p. Por ejemplo,
apila 4 p1 => 4|1|2|3|-|

apila :: a -> Pila a -> Pila a
apila x (P xs) = P (x:xs)

Implementación de las pilas mediante listas
I (cima p) es la cima de la pila p. Por ejemplo,

cima p1 == 1

cima :: Pila a -> a
cima (P (x:_)) = x
cima (P []) = error "cima de la pila vacia"

I (desapila p) es la pila obtenida suprimiendo la cima de la pila
p. Por ejemplo,
desapila p1 => 2|3|-

desapila :: Pila a -> Pila a
desapila (P []) = error "desapila la pila vacia"
desapila (P (_:xs)) = P xs

Implementación de las pilas mediante listas

I (esVacia p) se verifica si p es la pila vacía. Por ejemplo,
esVacia p1 == False
esVacia vacia == True

esVacia :: Pila a -> Bool
esVacia (P xs) = null xs

Importación de librerías

I Importación de la implementación de pilas que se desea
comprobar.

import PilaConTipoDeDatoAlgebraico
-- import PilaConListas

I Importación de las librerías de comprobación

import Test.QuickCheck
import Test.Framework
import Test.Framework.Providers.QuickCheck2

Generador de pilas
I genPila es un generador de pilas. Por ejemplo,

ghci> sample genPila
0|0|-
-6|4|-3|3|0|-
-
9|5|-1|-3|0|-8|-5|-7|2|-
...

genPila :: (Num a, Arbitrary a) => Gen (Pila a)
genPila = do xs <- listOf arbitrary

return (foldr apila vacia xs)

instance (Arbitrary a, Num a) => Arbitrary (Pila a) where
arbitrary = genPila

Especificación de las propiedades de pilas

I La cima de la pila que resulta de añadir x a la pila p es x.

prop_cima_apila :: Int -> Pila Int -> Bool
prop_cima_apila x p =

cima (apila x p) == x

I La pila que resulta de desapilar después de añadir cualquier
elemento a una pila p es p.

prop_desapila_apila :: Int -> Pila Int -> Bool
prop_desapila_apila x p =

desapila (apila x p) == p

Especificación de las propiedades de pila

I La pila vacía está vacía.

prop_vacia_esta_vacia :: Bool
prop_vacia_esta_vacia =

esVacia vacia

I La pila que resulta de añadir un elemento en un pila cualquiera
no es vacía.

prop_apila_no_es_vacia :: Int -> Pila Int -> Bool
prop_apila_no_es_vacia x p =

not (esVacia (apila x p))

Comprobación de las propiedades

Definición del procedimiento de comprobación

I compruebaPropiedades comprueba todas las propiedades con la
plataforma de verificación.

compruebaPropiedades =
defaultMain

[testGroup "Propiedades del TAD pilas"
[testProperty "P1" prop_cima_apila,
testProperty "P2" prop_desapila_apila,
testProperty "P3" prop_vacia_esta_vacia,
testProperty "P4" prop_apila_no_es_vacia]]

Comprobación de las propiedades

Comprobación de las propiedades de las pilas

ghci> compruebaPropiedades
Propiedades del TAD pilas:

P1: [OK, passed 100 tests]
P2: [OK, passed 100 tests]
P3: [OK, passed 100 tests]
P4: [OK, passed 100 tests]

Properties Total
Passed 4 4
Failed 0 0
Total 4 4

